query: SELECT ?item ?itemLabel ?count ?countBeschermd ?countPubliekdomein ?countBebekend ?beschermdPercentage ?publiekdomeinPercentage  ?bekendPercentage
  WITH {
    SELECT DISTINCT ?item ?human WHERE {
      ?human wdt:P31 wd:Q5 .
      ?human wdt:P6379 ?item.
      #?item wdt:P17 wd:Q55.# nl-instituut
    }
  } AS %humans
  WITH {
    SELECT DISTINCT ?human ?beschermd ?publiekdomein ?bekend WHERE {
      INCLUDE %humans.
      OPTIONAL {
        ?human wdt:P7763 wd:Q73555012
        BIND(1 AS ?beschermd)
      }
      OPTIONAL {
        ?human wdt:P7763 wd:Q71887839
        BIND(1 AS ?publiekdomein)
      }
      OPTIONAL {
        FILTER EXISTS
         {?human wdt:P7763 ?status
         }
        BIND(1 AS ?bekend)
      }
    }
  } AS %humansWithRep
  WITH {
    SELECT ?item (COUNT(?human) AS ?count) (SUM(?beschermd) AS ?countBeschermd) (CONCAT(str(ROUND(SUM(?beschermd)* 100 / (COUNT(?human)))), "% (", str(SUM(?beschermd)), " personen)") AS ?beschermdPercentage) (SUM(?publiekdomein) AS ?countPubliekdomein) (CONCAT(str(ROUND(SUM(?publiekdomein)* 100 / (COUNT(?human)))), "% (", str(SUM(?publiekdomein)), " personen)") AS ?publiekdomeinPercentage) (SUM(?bekend) AS ?countBekend) (CONCAT(str(ROUND(SUM(?bekend)* 100 / (COUNT(?human)))), "% (", str(SUM(?bekend)), " personen)") AS ?bekendPercentage) WHERE {
      INCLUDE %humans.
      INCLUDE %humansWithRep.
    }
    GROUP BY ?item
  } AS %items
  WHERE {
    INCLUDE %items.
    SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }
    FILTER (?count>1000)
  }